

MAC0444 - Sistemas Baseados em Conhecimento

Relatório do Projeto

Lincoln Oliveira - NUSP:
Marcos Castello - NUSP:
Matheus S. Freire - NUSP:
Natália Hitomi Koza - NUSP: 10698432

18 de Dezembro de 2021

Ontologia

Somente referenciamos a ontologia FOAF.

Conceitos	Descrição
Filme	subconceito de <code>foaf:Project</code>
Ator	subconceito de <code>foaf:Person</code>
Diretor	subconceito de <code>foaf:Person</code>
Personagem	subconceito de <code>foaf:Project</code>
Homem	subconceito de <code>foaf:Agent</code>
Mulher	subconceito de <code>foaf:Agent</code>

Propriedade	Descrição
atuaEm	subpropriedade de foaf:made inversa de atuadoPor propriedade assimétrica domínio: Ator range: Filme
atuadoPor	subpropriedade de foaf:maker
dirige	subpropriedade de foaf:made inversa de dirigidoPor domínio: Diretor range: Filme
dirigidoPor	subpropriedade de foaf:maker
interpreta	subpropriedade de foaf:made inversa de interpretadoPor domínio: Ator range: Personagem
interpretadoPor	subpropriedade de foaf:maker
temAnoDeLancamento	tipo: owl:FunctionalProperty domínio: Filme range: inteiro
temDuracao	domínio: Filme range: inteiro
temPrimeiroNome	subpropriedade de foaf:firstName tipo: owl:FunctionalProperty domínio: foaf:Person range: string
temSobrenome	subpropriedade de foaf:familyName tipo: owl:FunctionalProperty domínio: foaf:Person range: string
temTitulo	domínio: Filme range: string

Instâncias:

Consultas

Para todas as consultas os prefixos são:

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX owl: <http://www.w3.org/2002/07/owl#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
PREFIX foaf: <http://xmlns.com/foaf/0.1/>
PREFIX onto: <http://www.semanticweb.org/lincolnyuji/ontologies/2021/10/cinema-ontology#>
```

Query 1

Quais os títulos dos filmes que foram dirigidos pelo diretor **D**, em ordem lexicográfica? Com **D** = onto:ethan_coen:

```

SELECT ?TituloDoFilme
WHERE
{
  BIND (onto:ethan_coen AS ?D)
  ?D onto:dirige ?filme.
  ?filme onto:temTitulo ?TituloDoFilme.
}
ORDER BY ?TituloDoFilme

```

TituloDoFilme
"Barton Fink"
"Blood Simple"
"Fargo"
"Intolerable Cruelty"
"Miller's Crossing"
"O Brother, Where Art Thou?"
"Paris, je t'aime"
"Raising Arizona"
"The Hudsucker Proxy"
"The Ladykillers"
"The Man Who Wasn't There"

Query 2

Quais os primeiros nomes e últimos nomes dos atores que participaram do filme de título $\mathbf{F_t}$, em ordem lexicográfica de nome e sobrenome, com seus respectivos personagens? Com $\mathbf{F_t}$ = "American Beauty".

```

SELECT ?PrimeiroNome ?UltimoNome ?Personagem
WHERE
{
  BIND ("American Beauty" AS ?Ft)
  ?filme onto:temTitulo ?Ft.
  ?pessoa onto:atuaEm ?filme.
  ?pessoa onto:temPrimeiroNome ?PrimeiroNome.
  ?pessoa onto:temSobrenome ?UltimoNome.
  ?pessoa onto:interpreta ?Personagem.
}
ORDER BY ?PrimeiroNome ?UltimoNome

```

PrimeiroNome	UltimoNome	Personagem
"Alison"	"Faulk"	personagem_spartanette_1
"Allison"	"Janney"	personagem_barbara_fitts
"Amber"	"Smith"	personagem_christy_kane
"Annette"	"Bening"	personagem_carolyn_burnham
"Ara"	"Celi"	personagem_sale_house_woman_1
"Barry"	"Sherman"	personagem_brad_dupree
"Brenda"	"Wehle"	personagem_sale_house_woman_4
"Carolina"	"Lancaster"	personagem_spartanette_4
"Chekeshka"	"Putten"	personagem_spartanette_6
"Chelsea"	"Hertford"	personagem_playground_girl_2
"Chris"	"Cooper"	personagem_col_frank_fitts_usmc
"Chris"	"Cooper"	personagem_frank_booker
"Dennis"	"Anderson"	personagem_mr_smiley_s_manager
"Elaine"	"Kendall"	personagem_newscaster
"Emily"	"Zachary"	personagem_spartanette_7
"Erin"	"Strubbe"	personagem_young_jane_burnham
"Fort"	"Atkinson"	personagem_sale_house_man_2
"Heather"	"Sher"	personagem_playground_girl_1
"Joel"	"McCrary"	personagem_catering_boss
"John"	"Cho"	personagem_petey
"John"	"Cho"	personagem_sale_house_man_1
"Kent"	"Faulcon"	personagem_sale_house_man_3
"Kevin"	"Spacey"	personagem_lester_burnham
"Kevin"	"Spacey"	personagem_roger_verbal_kint
"Krista"	"Goodsitt"	personagem_spartanette_2
"Lily"	"Houtkin"	personagem_spartanette_3
"Lisa"	"Cloud"	personagem_sale_house_woman_5
"Marissa"	"Winokur"	personagem_mr_smiley_s_counter_girl
"Matthew"	"Kimbrough"	personagem_firing_range_attendant
"Mena"	"Suvari"	personagem_angela_hayes
"Nancy"	"Anderson"	personagem_spartanette_8
"Peter"	"Gallagher"	personagem_buddy_kane
"Peter"	"Gallagher"	personagem_vic_tenetta_party_singer
"Reshma"	"Gajjar"	personagem_spartanette_9
"Romana"	"Leah"	personagem_spartanette_5
"Sam"	"Robards"	personagem_jim_berkley
"Scott"	"Bakula"	personagem_jim_olmeyer
"Stephanie"	"Rizzo"	personagem_spartanette_10
"Sue"	"Casey"	personagem_sale_house_woman_2
"Thora"	"Birch"	personagem_enid
"Thora"	"Birch"	personagem_jane_burnham
"Wes"	"Bentley"	personagem_ricky_fitts

Query 3

Em quais filmes os atores **X** e **Y** atuaram juntos, com os respectivos diretores e anos de lançamento, do mais novo para o mais antigo?

Com **X** = onto:ben_gazzara e **Y** = onto:steve_buscemi

```
SELECT ?Filme ?Diretor ?Ano
```

```

WHERE
{
  BIND (onto:ben_gazzara AS ?X)
  BIND (onto:steve_buscemi AS ?Y)
  ?X onto:atuaEm ?Filme.
  ?Y onto:atuaEm ?Filme.
  FILTER (?X != ?Y).
  ?Diretor onto:dirige ?Filme.
  ?Filme onto:temAnoDeLancamento ?Ano.
}
ORDER BY DESC(?Ano)

```

Filme	Diretor	Ano
filme_paris_je_t_aime	alexander_payne	2006
filme_paris_je_t_aime	alfonso_cuar_n	2006
filme_paris_je_t_aime	anne_marie_mi_ville	2006
filme_paris_je_t_aime	christoffer_boe	2006
filme_paris_je_t_aime	daniela.thomas	2006
filme_paris_je_t_aime	ethan.coen	2006
filme_paris_je_t_aime	ettore_scola	2006
filme_paris_je_t_aime	fr_d_ric_auburtin	2006
filme_paris_je_t_aime	g_rard_depardieu	2006
filme_paris_je_t_aime	gus_van_sant	2006
filme_paris_je_t_aime	isabel.coixet	2006
filme_paris_je_t_aime	jean_luc_godard	2006
filme_paris_je_t_aime	joel.coen	2006
filme_paris_je_t_aime	nobuhiro_suwa	2006
filme_paris_je_t_aime	oliver_schmitz	2006
filme_paris_je_t_aime	olivier_assayas	2006
filme_paris_je_t_aime	richard_lagravenese	2006
filme_paris_je_t_aime	sylvain_chomet	2006
filme_paris_je_t_aime	tom_tykwer	2006
filme_paris_je_t_aime	vincenzo_natali	2006
filme_paris_je_t_aime	walter_salles	2006
filme_the_big_lebowski	joel.coen	1998

Query 4

Quais filmes do diretor do filme **F** possuem **X** ou **Y** como atores, com as respectivas durações em ordem crescente?

Com **F** = onto:filme_her, **X** = onto:amy_addams e **Y** = onto:tim_szczepanski.

```

SELECT ?Filme ?Duracao
WHERE
{
  BIND (onto:filme_her AS ?F)
  BIND (onto:amy_addams AS ?X)
  BIND (onto:tim_szczepanski AS ?Y)
  ?Diretor onto:dirige ?F.
  ?Filme onto:dirigidoPor ?Diretor.
}

```

```

    {?X onto:atuaEm ?Filme} UNION {?Y onto:atuaEm ?Filme}.
    ?Filme onto:temDuracao ?Duracao.
}
ORDER BY ?Duracao

```

Filme	Duracao
filme_torrance_rises	34
filme_her	126

Query 5

Quais pessoas atuaram em um filme e dirigiram algum filme (não necessariamente o mesmo filme, mas obrigatoriamente a mesma pessoa)?

```

SELECT DISTINCT ?Pessoa (SAMPLE(?filme1) AS ?FilmeAt) (SAMPLE(?filme2) AS ?FilmeDir)
WHERE
{
    ?Pessoa onto:atuaEm ?filme1.
    ?filme2 onto:dirigidoPor ?Pessoa.
}
GROUP BY ?Pessoa

```

Pessoa	FilmeAt	FilmeDir
harold_ramis	filme_ghostbusters	filme_groundhog_day
eleanor_coppola	filme_hearts_of_darkness_a_filmaker_s_apocalypse	filme_hearts_of_darkness_a_filmaker_s_apocalypse
woody_allen	filme_scoop	filme_scoop
george_lucas	filme_hearts_of_darkness_a_filmaker_s_apocalypse	filme_star_wars_episode_i_the_phantom_menace
ivan_reitman	filme_ghostbusters	filme_ghostbusters
ethan_coen	filme_down_from_the_mountain	filme_raising_arizona
roman_coppola	filme_star_wars_episode_i_the_phantom_menace	filme_cq
denzel_washington	filme_the_great_debaters	filme_the_great_debaters
va_g_rdos	filme_an_american_rhapsody	filme_an_american_rhapsody
g_rard_depardieu	filme_cq	filme_paris_je_t_aime
sofia_coppola	filme_inside_monkey_zetterland	filme_lost_in_translation
spike_jonze	filme_torrance_rises	filme_torrance_rises
francis_ford_coppola	filme_hearts_of_darkness_a_filmaker_s_apocalypse	filme_rumble_fish
sam_raimi	filme_spies_like_us	filme_crimewave
joel_coen	filme_spies_like_us	filme_fargo
eric_schaeffer	filme_if_lucy_fell	filme_if_lucy_fell
stephen_hillenburg	filme_the_spongebob_squarepants_movie	filme_the_spongebob_squarepants_movie
robert_redford	filme_the_horse_whisperer	filme_the_horse_whisperer

Query 6

Quais diretores dirigiram algum filme em um ano entre os anos N_1 e N_2 , em que os atores X e Y aparecem, do mais antigo para o mais novo?

```

SELECT DISTINCT ?Ano ?Diretor ?Filme
WHERE
{
    BIND (1901 AS ?N1)
    BIND (2000 AS ?N2)
}

```

```

BIND (onto:a_j_cook AS ?X)
BIND (onto:allen_stewart_coates AS ?Y)
?Filme onto:temAnoDeLancamento ?Ano.
FILTER (?Ano <= ?N2).
FILTER (?Ano >= ?N1).
?X onto:atuaEm ?Filme.
?Y onto:atuaEm ?Filme.
?Filme onto:dirigidoPor ?Diretor
}
ORDER BY ?Ano

```

Ano	Diretor	Filme
1999	sofia_coppola	filme_the_virgin_suicides

Query 7

Quais pares formados por uma atriz e um ator atuaram juntos em algum filme de duração entre M1 e M2?

```

SELECT ?Atriz ?Ator ?Filme ?Duracao
WHERE
{
  ?atriz rdf:type onto:Mulher.
  ?ator rdf:type onto:Homem.
  ?atriz onto:atuaEm ?filme.
  ?ator onto:atuaEm ?filme.
  ?filme onto:temDuracao ?duracao.
  FILTER (?duracao <= 120).
  FILTER (?duracao >= 80).
}
ORDER BY ?duracao

```

Query 8

Qual o diretor que mais dirigiu filmes em que aparece o ator de primeiro nome X_p e último nome X_u ?
Com X_p = Vincent, X_u = Coppola

```

SELECT ?Diretor
WHERE
{
  BIND ("Vincent" AS ?XP)
  BIND ("Coppola" AS ?XU)
  ?pessoa rdf:type foaf:Person.
  ?pessoa onto:atuaEm ?filme.
  ?pessoa onto:temPrimeiroNome ?XP.
  ?pessoa onto:temSobrenome ?XU.
  ?Diretor onto:dirige ?filme.
}
GROUP BY ?Diretor
ORDER BY DESC(COUNT(?filme))
LIMIT 1

```

Diretor
francis_ford_coppola

Query 9

Qual o filme mais antigo em que o ator **X** atuou, em que ano foi lançado e qual era seu personagem? Havendo empate, devolver todos. Com **X** = onto:alan_rachins.

```
SELECT ?Filme ?Ano ?Personagem
WHERE
{
  BIND (onto:alan_rachins AS ?X).
  ?X onto:atuaEm ?Filme.
  ?X onto:interpreta ?Personagem.
  ?Filme onto:temAnoDeLancamento ?anoD.
  FILTER (?anoD = ?Ano).
  {
    SELECT (MIN(?anoD) as ?Ano)
    WHERE
    {
      BIND (onto:alan_rachins AS ?X).
      ?X onto:atuaEm ?Filme.
      ?Filme onto:temAnoDeLancamento ?anoD.
    }
  }
}
```

Filme	Ano	Personagem
filme_north	1994	personagem_defense_attorney

Query 10

Qual o filme mais longo dirigido por **D**, e qual a sua duração? Havendo empate, devolver todos da mesma duração.

Para **D** = onto:ethan_coen.

```
SELECT ?Filme ?Duracao
WHERE
{
  BIND (onto:ethan_coen AS ?D)
  ?D onto:dirige ?Filme.
  ?Filme onto:temDuracao ?duracaoD.
  FILTER (?duracaoD = ?Duracao).
  {
    SELECT (MAX(?duracaoD) as ?Duracao)
    WHERE{
      BIND (onto:ethan_coen AS ?D)
      ?D onto:dirige ?Filme.
      ?Filme onto:temDuracao ?duracaoD.
    }
  }
}
```

Filme	Duracao
filme_paris_je_t_aime	120